

Глава 1

Векторная арифметика

Файл: Makefile

```
1 lab03.exe: lab03.obj
2 —→ ld lab03.obj -l kernel32 -o lab03.exe
3
4 lab03.obj: lab03.asm
5 —→ nasm lab03.asm -f win32
```

1.1 Система сборки СMake

Процесс сборки приложения из нескольких файлов может отличаться на разных компьютерах, например, могут использоваться разные пути для библиотек или использоваться разные компиляторы. Поэтому ручное написание Makefile является не желательным. Для создания Makefile обычно используют специальные утилиты, например, СMake, которую мы и рассмотрим в этой лабораторной работе.

1. Создайте рабочую директорию lab06 с репозиторием Mercurial.

2. Создайте файлы main.asm и СMakeLists.txt (обратите внимание на регистр символов в названии файлов, это будет важно при работе в ОС Linux):

Файл: main.asm

```
1 global _main ; вход в программу
2 extern _ExitProcess@4
3 extern _printf
4
5 section .text use32
6 _main:
7     push msg
8     call _printf
9     add esp, 4
10
11     xor eax, eax
12     push eax
13     call _ExitProcess@4
14 section .data
15 msg db "Hello World!!!", 13, 10, 0
```

Файл: СMakeLists.txt

```
1 cmake_minimum_required(VERSION 2.8.12)
2 project(lab06 C ASM_NASM)
3 add_executable(lab06 "main.asm")
```

```
4 set_target_properties(lab06 PROPERTIES LINKER_LANGUAGE C)
```

В первой строке файла CMakeLists.txt минимальная версия CMake, необходимая для сборки проекта. Во второй задаётся имя проекта и используемые языки программирования, хотя в нашей программе сейчас нет кода на языке Си, нам будут необходимы инструменты линковки. В третьей строке создаётся исполняемый файл lab06 и файлы с исходным кодом для его компиляции. В четвёртой строке мы указываем, что для линковки исполняемого файла надо использовать средства поставляемые с компилятором языка Си.

Далее скомпилируем нашу программу с использованием компиляторов MinGW и MS Visual Studio. Если раньше мы компилировали программы в той же директории, где хранились и исходные файлы, то теперь мы уйдём от этой вредной привычки.

3. Скомпилируйте приложение с использованием инструментов MinGW:

1. Создайте директорию build-mingw.
2. Откройте новый сеанс командной строки.
3. Перейдите в директорию lab06/build-mingw.
4. Внесите в начало переменной окружения PATH пути до NASM, CMake и MinGW.
5. Создайте Makefile для MinGW используя CMake:

```
.../lab06/build-mingw> cmake -G "MinGW Makefiles" ..
```

6. Посмотрите на созданный файл.
7. Соберите приложение (mingw32-make).
8. Проверьте работу программы.
4. В директории lab06 создайте файл .hgignore:

```
1 syntax: glob
2 build*
```

5. Зафиксируйте изменения с комментарием: «Init. Test compiling with MinGW tools.».

Теперь пришло время воспользоваться инструментами MS Visual Studio для сборки приложения. В дальнейшем мы будем их использовать для создания 64-разрядных приложений.

6. Попробуйте скомпилировать приложение с использованием MSVC:

1. В lab06 создайте директорию build-msvc.
2. Откройте новый сеанс командной строки.
3. Перейдите в директорию lab06/build-msvc.
4. Настройте окружения для работы с MSVC в x86-режиме:

```
...> call "{путь до MS VS}/VC/vcvarsall.bat" x86
```

5. Внесите в начало переменной окружения PATH пути до NASM и CMake.
6. Создайте Makefile для MSVC используя CMake:

```
.../lab06/build-msvc> cmake -G "NMake Makefiles" ..
```

7. Посмотрите на созданный файл.
8. Соберите приложение:

```
.../lab06/build-msvc> nmake
```

9. Что произошло? Можете ли вы самостоятельно сказать причину ошибки?

Как видно, линковщик не увидел реализацию двух функций. Эти функции находятся в среде выполнения языка Си, которая не была подключена к исполняемому файлу. Следовательно нашей задачей будет подключение необходимой библиотеки, но надо иметь в виду, что есть две реализации этой библиотеки: динамическая и статическая. По умолчанию используется динамическая библиотека, поэтому надо будет линковаться с библиотекой msvcrt.d.

7. Определите правило линковки с библиотекой msvcrt.d для исполняемого файла, при использовании инструментов MSVC в CMakeLists.txt:

```

1 cmake_minimum_required(VERSION 2.8.12)
2 project(lab06 C ASM_NASM)
3
4 add_executable(lab06 "main.asm")
5 set_target_properties(lab06 PROPERTIES LINKER_LANGUAGE C)
6
7 if(MSVC)
8     target_link_libraries(lab06 msvcrt)
9 endif()

```

8. Повторите сборку с помощью `make`.
9. Зафиксируйте изменения с комментарием «Fix for MSVC».

1.2 Библиотека функций

В этой части работы мы рассмотрим пример реализации вычислительной функции на Ассемблере и её вызов из программ написанных как на языке Ассемблер, так и на Си.

1.2.1 Соглашения о вызове `cdecl`

Наша задача написать 32-битную функцию для x86 процессоров, которая будет вызываться, в итоге, из программы написанной на языке Си. Для этого мы должны придерживаться правил передачи параметров и приёма результата определённых в этом языке. Такие правила называются «Соглашение о вызове», и в нашем случае нужно использовать соглашение `cdecl`.

Напомню общие правила `cdecl`. Параметры функции передаются через стек в направлении справа-налево. Выравнивание параметров: 4 байта. Результат функции передаются: в `EAX` для данных размером ≤ 4 байт; в паре `EAX:EDX` для данных размером ≤ 8 байт; в регистре `ST0` математического сопроцессора для действительных чисел; иначе вызывающая функция выделяет память для результата и помещает адрес в вершину стека (так называемый «скрытый параметр»), ожидая что в `EAX` будет реальный адрес, где сохранился результат. Вызывающая функция должна очистить стек самостоятельно. Подпрограмма может изменить состояние регистров `EAX`, `ECX` и `EDX`, значения остальных должны быть восстановлены подпрограммой. Стек математического сопроцессора должен быть пустым перед и после вызова.

1.2.2 Подпрограмма `Пи`

Как вам уже известно, константа `пи` вшита в математический сопроцессор, воспользуемся ей.

10. Создайте файл `mylib.asm`:

```

1 global _pi
2
3 section .text use32
4 _pi:
5     fldpi
6     ret

```

11. Теперь добавим новый файл в систему сборки `make`. Исправьте файл `CMakeLists.txt`:

```

1 ...
2 add_executable(lab06 "main.asm" "mylib.asm")
3 ...

```

12. Исправьте файл `main.asm`:

```

1 global _main
2 extern _ExitProcess@4
3 extern _printf
4 extern _pi
5
6 section .text use32
7 _main:
8     finit
9     call _pi
10    sub esp, 8
11    fstp qword [esp]
12    push msg
13    call _printf
14    add esp, 4+8
15
16    xor eax, eax
17    push eax
18    call _ExitProcess@4
19 section .data
20 msg db "Pi = %.20f", 13, 10, 0

```

13. Скомпилируйте программу и проверьте её работу. Сравните результат с эталонным значением числа Пи.

14. Добавьте файл `mylib.asm` под контроль версий.

15. Зафиксируйте изменения с комментарием «Added function pi.».

Теперь сравним точность представления чисел в `float` и `double`. Для этого сохраним результат функции дважды в разных форматах и выведем результат на экран используя `printf`. Но на втором этапе нас ожидает небольшая загвоздка: мы не можем передать переменную типа `float` напрямую в `printf`, необходимо преобразовать его в `double` (пункт 6.5.2.2-6 стандарта C99 [<http://www.open-std.org/JTC1/SC22/WG14/www/docs/n1256.pdf>]), т. к. она использует переменное количество параметров.

16. Добавьте вывод числа π в формате `float`.

```

1 ...
2     call _pi
3     sub esp, 16
4     fst dword [esp]
5     fstp qword [esp+8]
6     fld dword [esp]
7     fstp qword [esp]
8     push msg
9     call _printf
10    add esp, 4+16
11 ...
12 msg db "Pi(float) = %.20f", 13, 10
13     db "Pi(double) = %.20f", 13, 10, 0

```

17. Скомпилируйте программу и проверьте её работу. Какими командами происходит преобразование числа из `float` в `double`?

18. Зафиксируйте изменения с комментарием «Print float pi.».

1.2.3 Вызов функции из программы на языке Си

Теперь пришло время задействовать язык Си. Мы перепишем основную функцию нашей программы. Для этого корректно удалим файл `main.asm` и создадим новый `main.cpp`. Кроме этого нам понадобится заголовочный файл `mylib.h` для нашей библиотеки.

19. Средствами Mercurial удалите файл `main.asm`:

```
.../lab06> hg remove main.asm
```

20. Добавьте файлы main.c и mylib.h:

Файл: mylib.h

```
1 #ifndef _MYLIB_H_
2 #define _MYLIB_H_
3
4 double pi();
5
6 #endif // _MYLIB_H_
```

Файл: main.c

```
1 #include <stdio.h>
2
3 #include "mylib.h"
4
5 int main()
6 {
7     →double pi_d = pi();
8     →float pi_f = (float)pi_d;
9     →printf(
10    →→"Pi          ~ 3.14159265358979323846\n"
11    →→"Pi(float)   = %.20f\n"
12    →→"Pi(double)  = %.20f\n", pi_f, pi_d);
13    →return 0;
14 }
```

21. Подправим файл CMakeLists.txt:

```
1 cmake_minimum_required(VERSION 2.8.12)
2 project(lab06 C ASM_NASM)
3
4 add_executable(lab06 "main.c" "mylib.h" "mylib.asm")
5 set_target_properties(lab06 PROPERTIES LINKER_LANGUAGE C)
```

Стоит отметить, что теперь нет необходимости в явном указании необходимости линковки со средой выполнения Си, т. к. в нашем проекте присутствует файл с исходным кодом на Си.

22. Скомпилируйте программу и проверьте её работу. Сравните результат.

23. Добавьте новые файлы под контроль версий.

24. Зафиксируйте изменения с комментарием «Call pi from C.».

Найдите способ генерации кода на языке Ассемблер из исходного кода на языке Си используя компиляторы mingw32-gcc (из пакета MinGW) и cl (из MS VS). Сравните результат сгенерированного кода с вашей реализацией.